
Einleitung 

 Nachdem wir unsere erste Pivot Tabelle erstellt haben, führen wir die gleichen Schritte mit VBA 
durch. 

 
 
Erstellen einer Pivot Tabelle mit VBA 
 
Schritt 1: Ausgangstabelle erstellen 
Die Quelltabelle muss die folgenden Bedingungen erfüllen: 

 Tabellenüberschriften müssen vorhanden sein 

 Es dürfen keine Verbundzellen genutzt werden 

 Es muss ein zusammenhängender Zellbereich vorliegen (keine leeren Zeilen / Spalten / Zellen) 
 
Schritt 2: Leere Pivot Tabelle erstellen 

 In diesem Schritt definieren wir zunächst unser PivotCache, welches wir als Grundlage für das 
Erstellen der Pivot Tabelle benötigen. Das PivotCache speichert die Information, auf welche 
Quelltabelle wir uns beziehen.  

 Zudem erstellen wir noch ein neues Tabellenblatt, in welches wir dann unsere Pivot Tabelle 
einfügen werden. 

 Hinweise zur „Version“ findest Du weiter unten. 
 

Modul1 
Sub PivotTabelleErstellen()  
   
    'Variablen definieren  
    Dim pc As PivotCache    'Variable für das PivotCache 
    Dim pt As PivotTable    'Variable für die PivotTable     
 
    'Quelltabelle aktivieren 
    Worksheets("QUELLTABELLE").Activate 
 
    'PivotCache erstellen 
    Set pc = ActiveWorkbook.PivotCaches.Create( _  
 SourceType:=xlDatabase, _  
 SourceData:=Range("A1").CurrentRegion, _  
 Version:=xlPivotTableVersion14) 
 
    'Tabellenblatt für die PivotTabelle erstellen 
    Worksheets.Add 
    'Namen für das Tabellenblatt festlegen 

    ActiveSheet.Name = "PivotTable #" & ThisWorkbook.PivotCaches.Count + 1 

 
    'PivotTabelle erstellen 
    Set pt = pc.CreatePivotTable(TableDestination:=Range("A3"), _  
    TableName:= "Meine Tabelle " & ThisWorkbook.PivotCaches.Count + 1) 

 
End Sub 

Hinweis: Add-Methode 

 Für das Erstellen der Pivot Tabelle können wir auch alternativ die Add-Methode nutzen. 
    'PivotTabelle erstellen (weitere Möglichkeit) 
    Set pt = ActiveSheet.PivotTables.Add(PivotCache:=pc, tableDestination:=Range("A3")) 

 
 
Schritt 3: Darstellung der Pivot Tabelle 

 Nun ordnen wir unsere Tabellenüberschriften (PivotTable-Felder) dem gewünschten Feld (Filter, 
Spalten, Zeilen und Werte) zu. 

Modul1 
Sub PivotTabelleErstellen()  
   
    (...) 
     
    'PivotTabelle erstellen 
    Set pt = pc.CreatePivotTable(TableDestination:=Range("A3")) 
 
    'Felder zuordnen 

    With pt 

        'ZEILEN 

        .PivotFields("Datum").Orientation = xlRowField 

         

        'SPALTEN 

        .PivotFields("Produkt").Orientation = xlColumnField 

         

        'WERTE 

        .PivotFields("Betrag").Orientation = xlDataField 

         

        'FILTER 

        .PivotFields("Land").Orientation = xlPageField      

    End With 

 

End Sub 

 
 
Hinweis: Bezeichnung der Felder 

 Seit Excel 2010 haben die Felder neue Bezeichnungen. In der Tabelle siehst Du, welche 
Bezeichnungen jeweils in Excel und VBA genutzt werden. 

 

Alte Bezeichnung Neue Bezeichnung Bezeichnung in VBA 

Seitenfelder Filter PageField 

Spaltenfelder Spalten ColumnField 

Zeilenfelder Zeilen RowField 

Datenfelder Werte DataField 

Hinweis: PivotFields per Nummer ansprechen 

 Es besteht auch die Möglichkeit die PivotTable-Felder (PivotFields) über eine Zahl anzusprechen. 
Die Nummerierung erfolgt gemäß der Reihenfolge in der Quelltabelle (Standardeinstellung) oder 
alphabetisch (optionale Einstellung). Dies können wir beispielsweise nutzen, wenn wir mit einer 
Schleife arbeiten möchten. 

 
    'Felder über Zahl ansprechen 

    MsgBox pt.PivotFields(1) 'Datum 

    MsgBox pt.PivotFields(2) 'Produkt 

    MsgBox pt.PivotFields(3) 'Betrag 

    MsgBox pt.PivotFields(4) 'Land 

 
 
Hinweis: Version 

 Microsoft entwickelt Excel immer weiter und so kommen auch mit jeder neuen Excel Version neue 
Einstellungen, Funktionen und Anpassungen für die Pivot Tabellen. Dies gilt es zu berücksichtigen, 
wenn ein Code beispielsweise in Excel 2013 geschrieben wurde und auch unter anderen 
Versionen ausgeführt werden soll. Die Tabelle zeigt, welche Werte der Befehl xlPivotTableVersion 
annehmen kann.  

